```
/*Evaluation of FLuorometholone as Adjunctive MEdical Therapy for Trachomatous Trichiasis Surgery: The FLAME Randomized
Controlled Clinical Trial*/
/*purpose: Comparison of secondary efficacy outcomes between randomized treatment groups*/
/*Note: raw datasets in SAS library 'in' corresponds to the shared data file in xlsx format, their relationship are as below:
in.PRIMARY_END_W4 - week 4 record in PE
in.PRIMARY_END_M6 - month 6 record in PE
in.PRIMARY_END_M12 - month 12 record in PE
in.EYEEXAM_m12 - month 12 record in ME
in.SURGERY - SI
*/

/*set up SAS library, TLF path, TLF title, format catalog which will be used in later analysis*/
%include "setup.sas";
libname in "&_root_in";
libname fm "&_root_fm";
libname data "&_root_data";
%let TLF_path = &_root_output\tb4_secondary_endpoints.rtf;
%let TLF_title = Table 4:  Comparison of secondary efficacy outcomes between randomized treatment groups;

OPTIONS FMTSEARCH=(fm.fmsurgery fm.fmexamm6 fm.fmexamw4 fm.fmelig fm.fmprimary);
option mprint;

/*set up variable format for visualization*/
proc format;
        value gf  0 = "Placebo" 1 = "FML";
        value ynf 0 = "No" 1 = "Yes";
        value yno 1='No' 2='Yes';
```

```
value $eyelashf 'No trichiasis'='No trichiasis'
                                'Zero (epilating)'='Zero (epilating)'
                                '1-5'='1-5'
                                '6 or more'='6 or more';
value eyelasho  1='No trichiasis'
                                2='Zero (epilating)'
                                3='1-5'
                                4='6 or more';
value entropionpf 0='Absence' 1='Presence';
value entropionpo 1='Absence' 2='Presence';
value entropionf 0='None'
                                  1='Mild'
                                  2='Moderate'
                                  3='Severe'
                                  4='Complete';
value entropiono 1='None'
                                  2='Mild'
                                  3='Moderate'
                                  4='Severe'
                                  5='Complete';
value pval (default=8)
        low - <0.00095 = '<0.001'
        0.00095 - <0.0095 = [8.3]
        0.0095 - <0.045 = [8.2]
        0.045 - <0.0495 = [8.3]
        0.0495 - <0.04995 = [8.4]
        0.04995 - <0.05 = '~<0.05'
```

```
                0.05 = '0.05'
                0.05< - <0.05005 = '~>0.05'
                0.05005 - <0.0505 = [8.4]
                0.0505 - <0.055 = [8.3]
                0.055 - 0.99 = [8.2]
                0.99< - high = [8.2];
run;

/* combine covariate used for analysis;*/
data PRIMARY_END_W4;
set in.PRIMARY_END_W4;
run;

data PRIMARY_END_M6;
set in.PRIMARY_END_M6;
run;

data PRIMARY_END_M12;
set in.PRIMARY_END_M12;
run;

data EYEEXAM_m12;
set in.EYEEXAM_m12;
run;
options mprint;
/*reshape the dataset to be eye level*/
%proc_reshape(indata=WORK.PRIMARY_END_W4,memname="PRIMARY_END_W4",outdata=PRIMARY_END_W4_);
```

```
%proc_reshape(indata=WORK.PRIMARY_END_M6,memname="PRIMARY_END_M6",outdata=PRIMARY_END_M6_);
%proc_reshape(indata=WORK.PRIMARY_END_M12,memname="PRIMARY_END_M12",outdata=PRIMARY_END_M12_);
%proc_reshape(indata=WORK.EYEEXAM_M12,memname="EYEEXAM_M12",outdata=EYEEXAM_M12_);

data fas;
        set data.fas;
        if reye_eligibel=1 then do; eye='OD'; output; end;
        if leye_eligibel=1 then do; eye='OS'; output; end;
run;

proc sql;
create table data.tbl4_data as
select r.subjid, r.eye, r.treat,
          case when p1.peeye_3=1 or p2.peeye_3=1 or p3.peeye_3=1 then 1
                        when p1.peeye_3=0 or p2.peeye_3=0 or p3.peeye_3=0 then 0 end as surgery_ptt,
          case when mecornea_ul=0 and meepilation_ul<=0 then 'No trichiasis'
                        when mecornea_ul=0 and meepilation_ul in (1,2,3) then 'Zero (epilating)'
                        when 1<=mecornea_ul<=5 then '1-5'
                        when 6<=mecornea_ul then '6 or more' end as mecornea_ulc, mecornea_ul,
          case when memedial_ul=0 and meepilation_ul<=0 then 'No trichiasis'
                        when memedial_ul=0 and meepilation_ul in (1,2,3) then 'Zero (epilating)'
                        when 1<=memedial_ul<=5 then '1-5'
                        when 6<=memedial_ul then '6 or more' end as memedial_ulc, memedial_ul,
          case when melateral_ul=0 and meepilation_ul<=0 then 'No trichiasis'
                        when melateral_ul=0 and meepilation_ul in (1,2,3) then 'Zero (epilating)'
                        when 1<=melateral_ul<=5 then '1-5'
                        when 6<=melateral_ul then '6 or more' end as melateral_ulc, melateral_ul,
```

```
        case when sum(mecornea_ul,memedial_ul,melateral_ul)=0 and meepilation_ul<=0 then 'No trichiasis'
                when sum(mecornea_ul,memedial_ul,melateral_ul)=0 and meepilation_ul in (1,2,3) then 'Zero
(epilating)'
                when 1<=sum(mecornea_ul,memedial_ul,melateral_ul)<=5 then '1-5'
                when 6<=sum(mecornea_ul,memedial_ul,melateral_ul) then '6 or more' end as total_lashc,
        sum(mecornea_ul,memedial_ul,melateral_ul) as total_lash,
        case when meentropion_ul in (1,2,3,4) then 1
                when meentropion_ul=0 then 0 end as meentropion_ulp,meentropion_ul,
        /*to meet requirement of HIPAA, name (sisiname2) and certification (sisicert2) of surgeon is removed from the public
shared dataset*/
        sisicert2,sisiname2,
        /*for surgeon that only have a small number of surgery (<50), cluster them into 1 group*/
        case when sisicert2 in (/*surgeon license removed*/) then 'Other'
            else sisicert2 end as sisicert3
from fas as r
left join PRIMARY_END_W4_ as p1 on r.subjid=p1.subjid and r.eye=p1.eye
left join PRIMARY_END_M6_ as p2 on r.subjid=p2.subjid and r.eye=p2.eye
left join PRIMARY_END_M12_ as p3 on r.subjid=p3.subjid and r.eye=p3.eye
left join EYEEXAM_M12_ as e3 on r.subjid=e3.subjid and r.eye=e3.eye
left join in.surgery as s on r.subjid=s.subjid;
quit;

* descriptive statistics, hypothesis test and model building;
*%stat_char(input=data.tbl4_data,output=tab_1,unit=eye,variable=surgery_ptt,min=2,max=2,cohort=treat,ref=Placebo,questi
onlabel=,tab_order=1,value_label=ynf,order_label=yno,id=subjid,self_denom=1,N=1,gee_binomial=1,gee_multinomial=0,reve
rse=0,chi=0,fisher=0,surgeon=0,diffs=1,event=1);
```

```
%stat_char(input=data.tbl4_data,output=tab_1,unit=eye,variable=surgery_ptt,min=2,max=2,cohort=treat,ref=Placebo,questi
onlabel=,tab_order=1,value_label=ynf,order_label=yno,id=subjid,self_denom=1,N=1,gee_binomial=0,gee_multinomial=0,reve
rse=0,chi=0,fisher=2,surgeon=0,diffs=1,event=1,riskcol=2);
%stat_num(input=data.tbl4_data,output=tab_2_1,variable=mecornea_ul,cohort=treat,ref=Placebo,questionlabel=,tab_order
=2,id=subjid,pois=1,surgeon=1,surgeon_cluster=0,surg_bar=0,diffs=1);
%stat_char(input=data.tbl4_data,output=tab_2_2,unit=eye,variable=mecornea_ulc,min=1,max=4,cohort=treat,ref=Placebo,q
uestionlabel=Number of upper eyelid lashes touching the cornea at one
year,tab_order=2,value_label=$eyelashf,order_label=eyelasho,id=subjid,self_denom=1,N=1,gee_binomial=0,gee_multinomial
=0,reverse=0,chi=0,fisher=0);
data tab_2;
        format answerlabel $200.;
        set tab_2_1(in=a) tab_2_2(in=b);
        if a and index(answerlabel,'Mean (SD)')=0 then delete;
        else if a then do; answer_order=0.5; answerlabel='  Mean (SD)'; end;
run;
%stat_num(input=data.tbl4_data,output=tab_3_1,variable=memedial_ul,cohort=treat,ref=Placebo,questionlabel=,tab_order=
3,id=subjid,surgeon=1,surgeon_cluster=1,surg_bar=0,diffs=1,pois=1);
%stat_char(input=data.tbl4_data,output=tab_3_2,unit=eye,variable=memedial_ulc,min=1,max=4,cohort=treat,ref=Placebo,q
uestionlabel=Number of upper eyelid lashes touching the globe medial to the cornea at one
year,tab_order=3,value_label=$eyelashf,order_label=eyelasho,id=subjid,self_denom=1,N=1,gee_binomial=0,gee_multinomial
=0,reverse=0,chi=0,fisher=0);
data tab_3;
        format answerlabel $200.;
        set tab_3_1(in=a) tab_3_2(in=b);
        if a and index(answerlabel,'Mean (SD)')=0 then delete;
        else if a then do; answer_order=0.5; answerlabel='  Mean (SD)'; end;
run;
```

```
%stat_num(input=data.tbl4_data,output=tab_4_1,variable=melateral_ul,cohort=treat,ref=Placebo,questionlabel=,tab_order=
4,id=subjid,surgeon=1,surgeon_cluster=1,surg_bar=0,diffs=1,pois=1);
%stat_char(input=data.tbl4_data,output=tab_4_2,unit=eye,variable=melateral_ulc,min=1,max=4,cohort=treat,ref=Placebo,q
uestionlabel=Number of upper eyelid lashes touching the globe lateral to the cornea at 1
year,tab_order=4,value_label=$eyelashf,order_label=eyelasho,id=subjid,self_denom=1,N=1,gee_binomial=0,gee_multinomial
=0,reverse=0,chi=0,fisher=0);
data tab_4;
        format answerlabel $200.;
        set tab_4_1(in=a) tab_4_2(in=b);
        if a and index(answerlabel,'Mean (SD)')=0 then delete;
        else if a then do; answer_order=0.5; answerlabel='  Mean (SD)'; end;
run;
%stat_num(input=data.tbl4_data,output=tab_5_1,variable=total_lash,cohort=treat,ref=Placebo,questionlabel=,tab_order=5,i
d=subjid,pois=1,surgeon=1,surgeon_cluster=0,surg_bar=0,diffs=1);
%stat_char(input=data.tbl4_data,output=tab_5_2,unit=eye,variable=total_lashc,min=1,max=4,cohort=treat,ref=Placebo,ques
tionlabel=Total number of upper eyelid lashes touching the globe or the cornea at one
year,tab_order=5,value_label=$eyelashf,order_label=eyelasho,id=subjid,self_denom=1,N=1,gee_binomial=0,gee_multinomial
=0,reverse=0,chi=0,fisher=0);
data tab_5;
        format answerlabel $200.;
        set tab_5_1(in=a) tab_5_2(in=b);
        if a and index(answerlabel,'Mean (SD)')=0 then delete;
        else if a then do; answer_order=0.5; answerlabel='  Mean (SD)'; end;
run;
%stat_char(input=data.tbl4_data,output=tab_6_1,unit=eye,variable=meentropion_ulp,min=2,max=2,cohort=treat,ref=Placeb
o,questionlabel=Entropion in the upper eyelid at one
```

```
year,tab_order=6,value_label=entropionpf,order_label=entropionpo,id=subjid,self_denom=1,N=1,surgeon=1,surgeon_cluster
=0,surg_bar=0,diffs=1,event=1,gee_binomial=1,gee_multinomial=0,reverse=0,chi=0,fisher=0);
%stat_char(input=data.tbl4_data,output=tab_6_2,unit=eye,variable=meentropion_ul,min=1,max=5,cohort=treat,ref=Placebo,
questionlabel=Severity,tab_order=6,value_label=entropionf,order_label=entropiono,id=subjid,self_denom=1,N=1,gee_binomi
al=0,gee_multinomial=0,reverse=0,chi=0,fisher=0);
data tab_6;
        length answerlabel $200.;
        set tab_6_1(in=a) tab_6_2(in=b);
        if b then do; answerlabel='  '||answerlabel; answer_order=answer_order+3; end;
run;

data data.tbl4_result(drop=pvalue);
        length answerlabel cohort_0 cohort_1 diffs $200.;
        set tab_1(in=a) tab_2 tab_3 tab_4 tab_5 tab_6;
        if a then answerlabel='Reoperation for PTT during one year follow-up: Yes (%)';
        if pvalue^=. then pval=strip(put(pvalue,pval.)); else pval='';
run;
proc sort data=data.tbl4_result; by tab_order answer_order; run;

/*generate table*/
options nodate nonumber;
ods rtf file = "&TLF_path";
ods escapechar='~';
ods rtf text="~S={just=l font_size=10pt font_weight= bold} &TLF_title";
proc report data=data.tbl4_result nowd spanrows split='|' missing style(column)={background=white fontsize=9pt}

style(header)={background=white fontsize=9pt fontweight=medium};
```

```
        columns tab_order answer_order answerlabel cohort_0 cohort_1 diffs pval;
        define tab_order  / ' ' order order=internal noprint;
        define answer_order  / ' ' order order=internal noprint;
        define answerlabel / '' left style(column)={cellwidth=1.8in asis=ON} style(header)={just=l};
        define cohort_0 / "Placebo" center style(column)={cellwidth=1.4in};
        define cohort_1 / "FLuorometholone" center style(column)={cellwidth=1.4in};
        define diffs / 'Difference (95% CI)' center group style(column)={cellwidth=1.4in};
        define pval / 'P-value*' center group style(column)={cellwidth=1.4in};
        compute after tab_order;
                line '' ;
         endcomp ;
run ;
ods rtf text="~S={just=l font_size=9pt} *From generalized regression models adjusting for surgeon that account for inter-eye correlation except reoperation of PTT. for reoperation of PTT during one-year follow-up, fisher exact test is conducted due to small number of events";
ods rtf close;
```